Minesweeper

In this single-player game, you are clearing out a field of land mines by alternately uncovering mine-free spaces, and marking where you believe mines to be. Uncovered spaces may also be marked with a digit, indicating the number of adjacent mines. Use the numeric information to work out where the mines are located, and mark all of the mines to win the game.

Minesweeper was originally written by Robert Donner and Curt Johnson of Microsoft. It was converted to BASIC by Jim Allenspach.

```
                 MINESWEEPER
              CREATIVE COMPUTING
            MORRISTOWN NEW JERSEY
WOULD YOU LIKE INSTRUCTIONS (Y-N) ? N
GAME LEVEL (1=BEGINNER, 2=INTERMEDIATE, 3=EXPERT)? 1
#########
#########
#########
#########
#########
#########
#########
#########
#########

(U)NCOVER OR (M)ARK?  U
ENTER COORDINATES TO UNCOVER (ENTER 0 0 TO FINISH)? 4,4
WHEW -- NO MINE THERE!
#########
#########
#########
###1#####
#########
#########
#########
#########
#########

ENTER COORDINATES TO UNCOVER (ENTER 0 0 TO FINISH)? 3,3
WHEW -- NO MINE THERE!
..1######
..11222##
11....1##
#1111.1##
####1.111
#1111....
#1...111.
11..12#1.
....1##1.

ENTER COORDINATES TO UNCOVER (ENTER 0 0 TO FINISH)?  0,0
..1######
..11222##
11....1##
#1111.1##
####1.111
#1111....
#1...111.
11..12#1.
....1##1.

(U)NCOVER OR (M)ARK?  M
ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)? 4,1
COORDINATE FLAGGED
..1!#####
..11222##
11....1##
#1111.1##
####1.111
#1111....
#1...111.
11..12#1.
....1##1.

ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)? 1,4
COORDINATE FLAGGED
..1!#####
..11222##
11....1##
!1111.1##
####1.111
#1111....
#1...111.
11..12#1.
....1##1.

ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)?
```

Later in the Run

```
(U)NCOVER OR (M)ARK? M
ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)? 6,1
COORDINATE FLAGGED
..1!2!###
..1122221
11....111
!1111.1!1
111!1.111
11111....
!1...111.
11..12!1.
....1!21.

ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)? 7,1
COORDINATE FLAGGED
..1!2!!##
..1122221
11....111
!1111.1!1
111!1.111
11111....
!1...111.
11..12!1.
....1!21.

ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)? 9,1
COORDINATE FLAGGED
..1!2!!#!
..1122221
11....111
!1111.1!1
111!1.111
11111....
!1...111.
11..12!1.
....1!21.

ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)? 0,0
..1!2!!#!
..1122221
11....111
!1111.1!1
111!1.111
11111....
!1...111.
11..12!1.
....1!21.

(U)NCOVER OR (M)ARK? U
ENTER COORDINATES TO UNCOVER (ENTER 0 0 TO FINISH)? 8,1
```

```
WHEW -- NO MINE THERE!
..1!2!!2!
..1122221
11....111
!1111.1!1
111!1.111
11111....
!1...111.
11..12!1.
....1!21.

WHEW! YOU FOUND ALL OF THE MINES! CONGRATULATIONS!
PLAY AGAIN (Y-N)?
```

```
10 PRINT TAB(24);"MINESWEEPER"
20 PRINT TAB(20);"CREATIVE COMPUTING"
30 PRINT TAB(18);"MORRISTOWN NEW JERSEY"
40 INPUT "WOULD YOU LIKE INSTRUCTIONS (Y-N) "; Y$
50 IF Y$ <> "Y" THEN GOTO 150
55 PRINT
60 PRINT "CONGRATULATIONS! YOU HAVE BEEN RECRUITED BY THE ";
63 PRINT "US ARMY TO CLEAR LANDMINES"
65 PRINT "OUT OF ENEMY TERRITORY. YOU WILL BE IN CHARGE OF ";
67 PRINT "A LARGE PLOT OF LAND THAT"
70 PRINT "IS SPLIT UP INTO X/Y COORDINATES. WE KNOW HOW MANY ";
73 PRINT "MINES ARE IN THE LAND,"
75 PRINT "BUT NOT THEIR LOCATIONS. THAT'S *YOUR* JOB!"
80 PRINT
85 PRINT "FOR EACH TURN, YOU HAVE TWO POSSIBLE MOVES. YOU CAN ";
87 PRINT "EITHER ENTER A"
90 PRINT "COORDINATE THAT YOU WISH TO UNCOVER (IDEALLY ONE ";
93 PRINT "THAT YOU ARE PRETTY SURE"
95 PRINT "DOES NOT CONTAIN A MINE). IF YOU ARE NOT BLOWN UP, ";
97 PRINT "THAT LOCATION WILL BE"
100 PRINT "MARKED AS CLEARED, ALONG WITH ALL ";
103 PRINT "BORDERING COORDINATES THAT ARE MINE-"
105 PRINT "FREE. ANY CLEARED COORDINATES THAT ARE ADJACENT TO ";
107 PRINT "1 OR MORE MINES WILL"
110 PRINT "BE MARKED WITH THE NUMBER OF ADJACENT MINES."
115 PRINT
120 PRINT "THE OTHER MOVE IS TO MARK COORDINATES WITH ";
123 PRINT "FLAGS, INDICATING WHERE MINES ARE"
125 PRINT "(OR WHERE YOU THINK THEY ARE). YOU WILL NEED ";
127 PRINT "TO MARK ALL MINES TO WIN"
130 PRINT "THE GAME."
135 PRINT
140 PRINT "YOU WIN THE GAME BY MARKING ALL OF THE MINES, ";
143 PRINT "AND UNCOVERING ALL OF THE"
145 PRINT "NON-MINE COORDINATES."
147 PRINT
150 INPUT "GAME LEVEL (1=BEGINNER, 2=INTERMEDIATE, 3=EXPERT)"; L
155 ON L GOTO 165,170,175
160 PRINT "HUH?": GOTO 150
165 W=9: H=9: M=10: GOTO 180
170 W=16: H=16: M=40: GOTO 180
175 W=30: H=16: M=99
180 DIM A(W,H)
185 FOR N1=1 TO W
190 FOR N2=1 TO H
195 A(N1,N2)=0
200 NEXT N2
205 NEXT N1
210 FOR N1=1 TO M
215 M1=INT(RND(1)*W)+1
220 M2=INT(RND(1)*H)+1
225 IF A(M1,M2)=1 THEN GOTO 215
230 A(M1,M2)=1
240 NEXT N1
245 W1=0
300 REM THE MAIN LOOP
305 GOSUB 1000
307 IF W1=1 THEN GOTO 900
310 INPUT "(U)NCOVER OR (M)ARK"; A$
315 IF MID$(A$,1,1)="U" THEN GOTO 500
320 IF MID$(A$,1,1)="M" THEN GOTO 600
325 PRINT "HUH?": GOTO 310
500 REM UNCOVER COORDS
505 INPUT "ENTER COORDINATES TO UNCOVER (ENTER 0 0 TO FINISH)";
507 INPUT X,Y
510 IF X=0 AND Y=0 THEN GOTO 300
513 IF X<1 OR X>W OR Y<1 OR Y>H THEN PRINT "COORDINATE OUT OF
RANGE": GOTO 505
515 IF A(X,Y)=1 OR A(X,Y)=3 OR A(X,Y)=5 THEN GOTO 550
520 IF A(X,Y)>3 THEN PRINT "YOU'VE ALREADY UNCOVERED THAT COORDINATE":
GOTO 500
525 C1=X: R1=Y: GOSUB 2500
530 IF T1=0 THEN A(X,Y)=4 ELSE A(X,Y)=5+T1
535 PRINT "WHEW -- NO MINE THERE!"
540 GOSUB 2000
545 GOSUB 1000
546 IF W1=1 THEN GOTO 900
547 GOTO 505
550 REM *BOOM*
555 PRINT "*** BOOM *** -- YOU UNCOVERED A MINE!"
560 INPUT "DO YOU WANT TO SEE THE FINAL BOARD CONFIGURATION (Y-N)"; Y$
565 IF Y$="Y" THEN GOSUB 3000
570 INPUT "PLAY AGAIN (Y-N)"; Y$
575 IF Y$="Y" THEN GOTO 150
580 GOTO 999
600 REM MARK SOME COORDINATES
605 IF W1=1 THEN GOTO 900
607 INPUT "ENTER COORDINATES TO MARK AS A MINE (ENTER 0 0 TO FINISH)"; X,Y
610 IF X=0 AND Y=0 THEN GOTO 300
615 IF A(X,Y)=0 OR A(X,Y)=1 THEN A(X,Y)=A(X,Y)+2: PRINT "COORDINATE
FLAGGED": GOSUB 1000: GOTO 605
620 IF A(X,Y)=2 OR A(X,Y)=3 THEN A(X,Y)=A(X,Y)-2: PRINT "COORDINATE
UNFLAGGED": GOSUB 1000: GOTO 605
625 PRINT "THAT COORDINATE HAS ALREADY BEEN UNCOVERED": GOTO 605
900 REM THE USER WON!
905 PRINT "WHEW! YOU FOUND ALL OF THE MINES! CONGRATULATIONS!"
910 GOTO 570
999 END
1000 REM DISPLAY GAME BOARD
1003 U=0: C=0
1005 FOR N2=1 TO H
1010 FOR N1=1 TO W
1015 IF A(N1,N2)=0 OR A(N1,N2)=1 THEN PRINT "#";: C=C+1: GOTO 1040
1020 IF A(N1,N2)=2 OR A(N1,N2)=3 THEN PRINT "!";: GOTO 1040
1025 IF A(N1,N2)=4 THEN PRINT ".";: GOTO 1040
1030 IF A(N1,N2)=5 THEN PRINT "*";: GOTO 1040
1035 IF A(N1,N2)>5 THEN PRINT STR$(A(N1,N2) - 5);: GOTO 1040
1040 IF A(N1,N2)=3 THEN U=U+1
1050 NEXT N1
1055 PRINT
1060 NEXT N2
1065 PRINT
1070 IF U=M AND C=0 THEN W1=1
1090 RETURN
2000 REM TRY TO UNCOVER ALL UNCOVERABLE COORDINATES
2005 F=0
2010 FOR N1=1 TO W
2015 FOR N2=1 TO H
2020 IF A(N1,N2)<>4 THEN GOTO 2075
2025 FOR M1=-1 TO 1
2030 FOR M2=-1 TO 1
2035 IF M1=0 AND M2=0 THEN GOTO 2065
2040 IF N1+M1<1 OR N1+M1>W OR N2+M2<1 OR N2+M2>H THEN GOTO 2065
2045 IF A(N1+M1,N2+M2)<>0 AND A(N1+M1,N2+M2)<>2 THEN GOTO 2065
2050 C1=N1+M1: R1=N2+M2: GOSUB 2500
2055 IF T1=0 THEN A(N1+M1,N2+M2)=4: F=1
2060 IF T1>0 THEN A(N1+M1,N2+M2)=T1+5
2065 NEXT M2
2070 NEXT M1
2075 NEXT N2
2080 NEXT N1
2085 IF F=1 THEN GOTO 2000
2090 RETURN
2500 REM CALCULATE THE TOTAL NUMBER OF ADJACENT MINES
2505 REM IN A(R1,C1)
2510 REM TOTAL STORED IN T1
2515 T1=0
2520 FOR N3=-1 TO 1
2525 FOR M3=-1 TO 1
2530 IF C1+N3<1 OR C1+N3>W OR R1+M3<1 OR R1+M3>H THEN GOTO 2540
2535 IF A(C1+N3,R1+M3)=1 OR A(C1+N3,R1+M3)=3 OR A(C1+N3,R1+M3)=5 THEN
T1=T1+1
2540 NEXT M3
2545 NEXT N3
2550 RETURN
3000 REM CONVERT THE BOARD INTO SIMPLE UNCOVERED EMPTY/MINE CELLS
3005 FOR N1=1 TO W
3010 FOR M1=1 TO H
3015 IF A(N1,M1)=0 OR A(N1,M1)=2 OR A(N1,M1)>5 THEN A(N1,M1)=4
3020 IF A(N1,M1)=1 OR A(N1,M1)=3 THEN A(N1,M1)=5
3025 NEXT M1
3030 NEXT N1
3035 GOSUB 1000
3040 RETURN
```